Adding a multi-line console and cheat code system to your game.

While working on a rewrite of the templates for a commercial game I have in development I saw the following code in MENU.WDL:

////////////////////////////////////////////////////////////////////////////

// For debugging purposes, use the EXECUTE instruction

// to type in WDL instructions during gameplay, like at a console.

// You can examine skill values through "TO_STRING look,skill;"

STRING exec_buffer
// just an 80-char string

"                                                                    ";

STRING look "         "; // to see skills via TO_STRING;

TEXT console_txt

{


POS_X 4;


LAYER
10;


FONT 
standard_font;


STRINGS 3;

IFDEF CONSOLE_MODE_2;


STRING "Enter instructions below, abort with [ESC]:";

IFELSE;


STRING "Enter instructions below:";

ENDIF;


STRING exec_buffer;


STRING look;

}

function console()

{


if(console_txt.VISIBLE == ON) { RETURN; }
//already running


console_txt.POS_Y = SCREEN_SIZE.Y - 60;


console_txt.VISIBLE = ON;


while(console_txt.VISIBLE == ON)


{



INKEY
exec_buffer;



if(RESULT == 13)



{




EXECUTE exec_buffer;

IFDEF CONSOLE_MODE_2;



}



else



{




console_txt.VISIBLE = OFF;



}

IFelse;



}



console_txt.VISIBLE = OFF;

ENDIF;


}

}

// Implementing a scrolling console, instead of a 1-line one,

// is left as an exercise to the reader...

Knowing that I wanted to extend this functionality anyway I looked at how other commercial games implemented this functionality and chose the “Return to Castle Wolfenstein” as the model that I wanted to use. We should first look at the last comment in the original script. This looks like a challenge, and I am always up for a challenge.  

The first thing that we need to do is define what it is that we want the script to do. This is an important step because we need a target to know where we are going and when we have arrived. For my game the console should:

1) Have 24 lines of text.

2) Have a function that can be called from other modules to write information to the console even if it is not visible at the time.

3) Auto scroll the console on write.

4) Be able to tell WDL commands from “cheat codes”.

5) Slide down and slide back up on activation and deactivation. 

6) Display an error message if the use enters an invalid command.

We should be able to do this with a minimum of functions. The prototypes of the functions needed are:

function _console_init();

function _scroll_console();

function _write_console(new_text);

function _parse_exec_buffer();

function _console();

In order to make a console that covers the game we will need a background panel and a text area on top of this panel. The text area will need 24 lines. This is a good place to talk about global variables and how not to use them. The text are will need 24 strings and a typical game programmer would make a definition for this text area like this.

string _disp1[80];

…

string _disp24[80];

text _console_txt 

{

  pos_x 4;

  layer 12;

  font standard_font;

  strings 24;

  string _disp1;

  …

  string _disp24;

}

This would create 24 global variables that hold the strings for the console. It would be better if these strings did not have global exposure. Defining the strings in the text area as follows can do this.

text _console_txt

{

  pos_x 4;

  layer
12;

  font 
standard_font;

  strings 24;

  string "                                                         ";

…

  string "                                                         ";

}

The strings are now defined locally to the text area. These strings can be access by using the .string[x] construct. Now we need our global variables that truly need to be global. Well actually only one of them needs to be global but because string and string* can not be used to create local string variables we need to create this type of variable globally.  We will need one string pointer to point to the string in our text area that is used for input and three strings for use as buffers for parsing and one flag variable to indicate that the console system has been initialized.

//temp string buffers used for !print command

string* key_buff;

string _tmp_buffer[80];

string _exec_buffer[80];

string _exec_buffer2[80];

//variable used to detect if we have called _console_init

var    _console_init_done = 0;

We will also need, as stated earlier a background panel. This panel and our font are the graphics resources we need for this functionality. They would be defined as follows:

//graphic resources

font standard_font,<ackfont.pcx>,6,9;

bmap _console_pcx = <console.pcx>;

//the console termainl screen background panel

panel _console_bgnd

{

  layer 11;

  bmap = _console_pcx;

  flags refresh,d3d;

}

We now have the definitions done for our new text console. It is important to note that the string length defined in the text area by the quoted strings is the maximum size that for each line and that they must all be the same length.  It should also be if we make a string zero length then the engine will not display that line. One other complication is that if the game is being restored then the strings in the text area will be restored from the game load. It is important that we clear the console area  after the game load. Because of the asynchronous nature of the game engine we need to add a wait to _console_init() function if we have not initialized the console system before. So with all this in mind the _console_init() function could look like this:

//this function will copy the blank line to each line in the display

//to clear old text from the display. This function sets the flag

//variable _console_init_done to indicate that the console is ready to

//use. 

function _console_init()

{

  //make sure that we wait because if we are called from

  //the main function all the variables may not be loaded. 

  if (_console_init_done == 0) { wait(2); }

  str_cpy(_console_txt.string[0],"");

…

  str_cpy(_console_txt.string[22],"");

  str_cpy(_console_txt.string[23]," ");

  _console_init_done = 1;

}

As can be seen from the previous code we are using the string[] array from the text object.  The obvious optimization of this code would be to use a loop to iterate through the strings. The previous code would be a common situation that a beginner would create. While it will function correct the following code is much clearer and more maintainable that the previous code.  A return(0) was also added to give the function a predictable return value.

//this function will copy the blank line to each line in the display

//to clear old text from the display. This function sets the flag

//variable _console_init_done to indicate that the console is ready to

//use. 

function _console_init()

{

  var _index;

  //make sure that we wait because if we are called from

  //the main function all the variables may not be loaded. 

  if (_console_init_done == 0) { wait(2); }

  _index = 0;

  while (_index < 23)

  {

    str_cpy(_console_txt.string[_index],"");

    _index += 1;

  }

  str_cpy(_console_txt.string[23]," ");

  _console_init_done = 1;

  return(0);

}

We can scroll the text by simply copying the text from the line below to the current line in the loop. Now the function to scroll the screen will be a simple matter of making a few small changes to this previous code. 

//this function will copy each line to the line above itself. This

//is how we implement the scrolling of the text. 

function _scroll_con()

{

  var _index;

  if (_console_init_done == 0) { _console_init(); wait(4); }

  _index = 0;

  while (_index < 23)

  {

    str_cpy(_console_txt.string[_index],_console_txt.string[_index + 1]);

    _index += 1;

  }

  str_cpy(_console_txt.string[23]," ");

  return(0);

}  

Note the line that starts with “if  (_console_init_done == 0)…”. Here we call _console_init() and then wait for four frames. I _console_init() if we have not finished an initialization then we will wait for two frames just in case the game load is not complete. By waiting for four frames in the calling function we guarantee that we will run after the initialization function is run.

Now we come to the function that will write to our console. This function should also check for initialization and should scroll the text are when done so that the line we use for input is empty. The code for this function is as follows:

//This function will copy the new_text value to the last line in the display

//and then scroll the text display. If _console_init_done has not been set then

//_console_init will be called. The wait(4) instruction is used to keep 

//synchonization with the _console_init function. 

function _write_console(new_text)

{

  if (_console_init_done == 0) { _console_init(); wait(4); }

  str_cpy(_console_txt.string[23],new_text);

  _scroll_con();

  return(0);

}

We are now ready to consider the command parser that we will use to create cheat codes for our game. By integrating this functionality into the console we provide a way to ad many debugging and cheat codes to our game without creating a situation that could cause errors or unpredictable results in other parts of the game. As an example of the kind of error that could happen if we were to use a simple state machine to detect a sequence of  key presses and our default movement keys had been changed to include keys from this sequence we would move our avatar while entering the sequence. This could result in our avatar falling over the edge of a cliff. This leaves us we the only options of not allowing key reassignment or testing that if we reassign keys that we can still enter codes at any place in the game and not do unexpected actions. The solution presented here looks for commands that start with “!” as special commands that we will then parse to determine the action needed. This is done using the str_stri string function and looking at the location that is returned. To make this parser simple we have made it a requirement that the command at the start of the line with no white space.  I have also removed the code for doing most of the actions as this would be game and template specific. I also return 1 if there was a command found that can be executed and 0 if no command was found to execute. The following code is the example parser:

//This function will parse the input buffer for special commands and then

//execute the code needed for each command. 

function _parse_exec_buffer()

{

  if (str_stri(_exec_buffer,"god") == 2) {

    _scroll_con();

    _write_console("God Mode");

    return(1);

  }

  if (str_stri(_exec_buffer,"give") == 2) {

    if (str_stri(_exec_buffer,"ammo") >= 7) {

      _scroll_con();

      _write_console("All ammo given");

      return(1);

    }

    if (str_stri(_exec_buffer,"health") >= 7) {

      _scroll_con();

      _write_console("Health restored");

      return(1);

    }

    if (str_stri(_exec_buffer,"keys") >= 7) {

      _scroll_con();

      _write_console("All keys given");

      return(1);

    }

    if (str_stri(_exec_buffer,"weapons") >= 7) {

      _scroll_con();

      _write_console("All weapons given");

      return(1);

    }

    if (str_stri(_exec_buffer,"all") >= 7) {

      _scroll_con();

      _write_console("All ammo, keys, weapons given, health restored");

      return(1);

    }

  }

  if (str_stri(_exec_buffer,"cls") == 2) {

    _console_init();

    return(1);

  }

  if (str_stri(_exec_buffer,"print") == 2) {

    str_cpy(_tmp_buffer,_exec_buffer);

    str_clip(_tmp_buffer,6);

    str_cpy(_exec_buffer2,"str_for_num(_tmp_buffer,");

    str_cat(_exec_buffer2,_tmp_buffer);

    str_cat(_exec_buffer2,");");

    str_cpy(_exec_buffer,_tmp_buffer);

    str_cat(_exec_buffer," = ");

    str_cpy(_tmp_buffer,"");

    execute(_exec_buffer2);

    str_cat(_exec_buffer,_tmp_buffer);

    _scroll_con();

    _write_console(_exec_buffer);

    return(1);

  }

  return(0);

}

Now we need a function to tie this all together and add the slide down and slide up functionality. Because we did not set a y location on either the panel or the text area we can change this in a simple loop. We can also use the time function to make the sliding the same on systems with different speeds. The key feature of this code is that we use the string pointer key_buff to point to the last line of the text area.  This is because the C-script compiler complains that “inkey _console_txt.string[23];” is an error. we then get the result of the inkey call and store it in a local variable so that we can test the exit code more than once. If the user presses enter then we check for “!” as the first character. If we find “!” is the first character then we parse the buffer otherwise we simple execute the buffer. If the user presses the TAB key we scroll the text area and background back up and then hide them from view. The code to the _console() function is as follows:

function _console()

{

  var res;

  if (_console_init_done == 0) { _console_init(); wait(4); }

  if(_console_txt.visible == on) { return; }
//already running

  _write_console("Enter commands. Press TAB to exit.");

  _console_txt.pos_y =  -256;

  _console_txt.visible = on;

  _console_bgnd.pos_y = -256;

  _console_bgnd.visible = on;

  while(_console_txt.pos_y < _console_txt.char_y) {

    _console_txt.pos_y += 10*time;

    _console_bgnd.pos_y +=10*time;

     if(_console_txt.pos_y > _console_txt.char_y) {

        _console_txt.pos_y = _console_txt.char_y;

        _console_bgnd.pos_y = 0;

     }

     wait(1);

  }

  while(_console_txt.visible == on)

  {

    key_buff = _console_txt.string[23];

    inkey key_buff;

    res = result;

    str_cpy(_exec_buffer,_console_txt.string[23]);

    if(res == 13)

    {

      if (str_stri(_exec_buffer,"!") == 1) 

      {

        if (!_parse_exec_buffer())

        {

          _scroll_con();

          _write_console("Error: invalid command");

        }

      } else {

        _scroll_con(); 

        execute(_exec_buffer);

      }

    }

    else

    { 

      if(res==9) 

      {

        while(_console_txt.pos_y > -_console_txt.char_y * 25) 

        {

          _console_txt.pos_y -= 10*time;

          _console_bgnd.pos_y -=10*time;

          if(_console_txt.pos_y < -(_console_txt.char_y * 25)) 

          {

            _console_txt.pos_y = -(_console_txt.char_y * 25);

      
    _console_bgnd.pos_y = -256;

      
  }

      
  wait(1);

      
}

        _console_txt.visible = off;

        _console_bgnd.visible = off;

        bmap_purge(_console_pcx);

      }

    }

  }

}

The only thing left is to point the tab key to this function and that is a simple on_tab assignment.

on_tab _console;

With this new console you should be able to add calls to _write_console() to various parts of your game to indicate order of execution and display critical values. In conclusion we have covered many areas of design and implementation in this tutorial and the fact that there is usually more that one way to get the desired results. I will be presenting a new movement/input/key reassignment module next. Look for that tutorial sometime later this year.

